
Resumen

Entre las caracteristicas de los programas que determinan su calidad, hay aspectos de fondo
(correccion y eficiencia) y de forma (disefio, estilo y documentacién). Cada vez mas, se reconoce la
importancia de estos tltimos: repercuten directamente en los primeros, y también tienen interés por
si mismos, ya que el mantenimiento y reutilizaciéon de los programas depende basicamente de ellos.
En estas péaginas se recogen brevemente distintos aspectos relacionados con el estilo en programacion
imperativa en Pascal.

1 Introduccién

El modelo imperativo es actualmente el mas frecuentemente adoptado en cursos de introduccién a
la programaciéon. Aunque el lenguaje Pascal no es desdeluego el tinico en estos cursos, ocupa un lugar
insustituible hoy por hoy en un buen nimero de plantes de estudio universitarios.

Es corriente plantear muchas de estas asignaturas partiendo del objetivo general de la ensenanza de
la programacion:

“Capacitar a los alumnos para construir metédicamente programas correctos, faciles de man-
tener y reutilizar y eficientes”

Sin entrar ahora en muchos detalles sobre lo ambicioso que es este objetivo, si podemos afirmar
que la adquisicion de unos buenos hébitos en el desarrollo de los programas incide directamente en la
correccién, mantenimiento y reutilizacion. El proceso de desarrollo de los programas ha de ser metédico
(diseno descendente y refinamientos sucesivos), y el programa final debera responder a distintos criterios
de calidad:

e La correccion es desde luego el objetivo principal de un programa: Un programa debe verificar
su especificaciéon. Por supuesto, no vamos a entrar aqui en este asunto. Pero si senalamos que
los aspectos que siguen (disefio, estilo y documentacién) contribuyen al desarrollo de programas
correctos, aunque se consideren por separado.

¢ Fl diseno atiende a distintos aspectos seménticos y metodoldgicos, como son la organizacion de los
programas, el buen desarrollo de las distintas estructuras, subprogramas tipos de datos y TADs.
La finalidad de un buen disefio es lograr programas que, ademés de correctos, sean mantenibles y
reutilizables.

e El estilo y la documentacion atienden a aspectos mas formales como son el buen uso de los
identificadores y comentarios, o la presentacién del texto del programa. La finalidad es que los
programas sean legibles con facilidad y naturalidad.

e La eficiencia es otro criterio de la calidad de los programas, pero en este folletito no lo tenemos
en cuenta.

Ya se ve que el diseno, el estilo y la documentacién no son los tinicos objetivos importantes. E incluso
su importancia puede considerarse secundaria en comparacién con otros aspectos de la calidad de los
programas. Pero, precisamente por ello, se ignora o desprecia con frecuencia el papel que desempenan
aspectos formales, como son la disposicién, presentacion, elecciéon de identificadores, etc.

Ya Gries resalto la importacia de estos aspectos [2], y subray6 su papel, no como adornos que decoran
un programa desarrollado, sino como habitos que todo programador debe practicar durante el proceso de
la programacion.

El objeto de estas paginas es justamente recoger estos aspectos.

2 Criterios generales

Un idea late bajo todo esto: que los programas han de ser ejecutados por méquinas, pero han de
ser leidos y mantenidos por humanos. Es necesario por tanto facilitar la lectura de los programas y, por
consiguiente, la comprensién de la logica del programa, de qué hace y por qué. Para lograrlo, recogemos
los siguientes objetivos para un programa, estrechamente relacionados entre si:

e Programas claros y naturales. La finalidad es que un programa sea facil de leer y entender, por
distintos programadores, y no sé6lo por el que lo escribi6. Como ideal, se puede plantear asi: “que
un programa se lea con la misma facilidad y naturalidad que si fuera una buena novela”.

La razon es facilitar el razonamiento sobre la correccién, su mantenimiento y su reutilizacion.

e Programas autoexplicativos: La idea bésica es que mantener dos textos distintos para un mismo
programa (el codigo y la documentacion) propicia los errores, debido a la dificultad de mantenerlos
consistentes [2].

Por tanto, conviene integrar el programa y la explicacién del mismo en un mismo documento. El
gran énfasis que se pone en la documentaciéon del programa recibe el nombre de programacion
letrada [3].

e Programas con un estilo homogéneo y consistente. Estas caracteristicas contribuyen a facilitar
la lectura de los programas.

Veremos seguidamente una serie de recomendaciones concretas con las que lograr programas que se
adapten a los criterios anteriores.

3 Recomendaciones

3.1 Sobre las declaraciones

e La primeraregla: usar identificadores significativos en el contexto del problema que se esta resolvien-
do. Los identificadores impronunciables deben evitarse. No usar abreviaturas indiscriminadamente:
solo las consagradas (EEUU, pta, kg, etc.), que se entiendan por si mismas.

e La tnica razon para limitar el tamano de los identificadores es evitar el riesgo de colisiones entre
nombres cuando se emplean compiladores que sélo reconocen una parte y truncan el resto.

e Para los datos constantes o variables, son apropiados los sustantivos, salvo con los de tipo booleano,
que se bautizan bien mediante afirmaciones (que pueden ser ciertas o falsas):

CONST
PorCiento = 0.01;
VAR
sueldo : real;
distancia : real;
inicial : char;
monedero : integer;
cuentaBanco : integer;

quedanFondosEnCaja: boolean;
Obsérvese la naturalidad de su uso en el texto del programa:

sueldo:= sueldo * (1 + 14 * PorCiento);
cuentaBanco:= cuentaBanco + sueldo; {cobro, a primeros de mes}

e Los procedimientos se pueden nombrar adecuadamente con identificadores que representan verbos
en infinitivo:

SacarCantidadMetalico(cuentaBanco, monedero, 10000)
También es adecuado usar el imperativo,

SacaCantidadMetalico(cuentaBanco, monedero, 10000)

reflejando el espiritu del modelo (imperativo) de programacion de Pascal.
Cualquiera de las dos opciones es vlida, pero no deben mezclarse ambas.

e Las funciones deben nombrarse segin sea el tipo de datos devuelto; de este modo, su utilizacién se

3.2

adecuara al mismo (sustantivo o proposicion).
inicial:= Primeraletra(nombre) ;
amortizacion:= Mensualidad(deuda, interesNominal, plazo);

REPEAT

UNTIL NOT HayFondosDisponibles(hucha, cuentaBanco)

Sobre el uso de constantes y variables

e Es habitual el uso erréneo de variables en un programa, por dos motivos frecuentes: uno, la presencia

de variables sin asignar inicialmente, y dos, variables asignadas inicialmente sin motivo.

e El texto del programa debe evitar el uso de valores numeéricos literales. La primera razén es que

3.3

dificulta el mantenimiento, porque cambiar un valor literal debe hacerse minuciosamente rastreando
a lo largo de todo el programa todas sus apariciones, més las de los valores que dependan de él. Otra
razén es que el papel de los valores bien identificados (como PI por ejemplo) se interpreta mejor
que sus correspondientes literales (como 0.3141592E01 por ejemplo) cuando se lee un programa.

Sobre las expresiones

e Un habito frecuente en expresiones booleanas es compararlas con las constantes true y false:

3.4

IF exzprbool; = true THEN
WHILE ezprbooly = false DO

Estas expresiones se debian haber escrito mejor simplemente asi:

IF exzprbool; THEN
WHILE NOT ezprbools DO

Sobre las instrucciones estructuradas

e Nunca usar la instrucciéon GOTO.

e Al desarrollar las instrucciones de selecciéon (condicionales y por casos), se deben contemplar todas

las posibilidades de la expresion selectora. Pero esto no significa que se tengan que incluir siempre
todas las ramas (por ejemplo, los ifs sin else son lo mas normal del mundo), sino que se deben
haber previsto todas las situaciones posibles.

e La eleccion de los bucles debe responder a su cometido en el programa: ntimero de iteraciones fijo

(FOR), conocido a priori, o terminaciéon condicionada; ninguna vuelta o mas (WHILE) o una vuelta
como minimo (REPEAT).

e En los bucles FOR, nunca se deben modificar la variable-indice ni el limite superior.

3.5

Tras un bucle FOR, el valor de la variable-indice es desconocido; por lo tanto, no se debe usar ese
hipotético valor.

Sobre los subprogramas

e El uso de variables globales debe considerarse prohibido. En efecto, un subprograma que usa o

modifica variables globales no est& autocontenido. Por consiguiente, esta practica ocasiona muchos
errores, dificiles de localizar.

Un remedio ideal para evitar este problema es declarar todas las variables de un programa después
de los subprogramas.

3.6

Debe evitarse el uso de subprogramas con muchos parametros, ya que esto los hace dificiles de leer
y usar. Este defecto indica usualmente un diseno pobre. Con frecuencia, este defecto se debe a usar
varios pardmetros que, en realidad, representan un solo dato, convenientemente agrupadas. Por
ejemplo, en vez de escribir

AvanzarFecha(numDia, diaSemana, numMes, anyo)
podria hacerse la llamada siguiente:
AvanzarFecha(dia)

donde la variable dia es un registro que contiene todos los datos de una fecha concreta: el nimero
de dia, el dia de la semana, el nimero del mes y el ano.

Debe evitarse el uso de subprogramas largos (méas de dos paginas se considera excesivamente largo),
que también son dificiles de comprender y depurar. Este defecto indica también un diseno pobre.

Los parametros de las funciones deberan ser siempre datos de entrada, ya sea mediante el mecanismo
de paso de pardmetros por valor o mediante el calificador CONST, que se puede aplicar a datos de
gran tamano para mejorar la eficiencia; en este ultimo caso, estos parametros nunca se podrin
modificar en el cuerpo de la funcién.

Las funciones no deben efectuar operaciones de lectura ni escritura, ya que estas operaciones produ-
cen efectos laterales sobre las variables “entrada” y “salida” de datos. En resumen, todo subprograma
que lee o escribe debe ser un procedimiento.

Por otra parte, los procedimientos que calculan nunca deben leer ni escribir, y los que escriben o
leen nunca deben calcular.

La forma en que se expresa el resultado de una funcién en Pascal (la asignacion a la seudovariable
de la funcion) es un error de diseno sustancial. Por consiguiente, se recomienta diseflar programas
que recaigan lo minimo posible en este diseno. Un modo usual es definir una variable resultado,
local a la funcién, que almacena el resultado, y que se asigna al nombre de la funcion en la altima
linea. Otra solucién es usar la variable result, que existe en varios dialectos de Pascal.

El uso de una variable intermedia es el tinico modo de facilitar la depuracion.

En los lenguajes que permitan definir subprogramas sobrecargados (por ejemplo, freePascal permite
definir operadores sobrecargados), todos los ejemplares deben responder al mismo proposito; de otro
modo, se ocasionaria confusion. Por ejemplo, es posible sobrecargar el operador ‘+’ para trabajar
con nuimeros complejos o con vectores, pero solo es adecuado hacerlo con el cometido de sumar esos
tipos de datos.

Sobre los tipos de datos

Que reflejen de forma natural las entidades del problema: enumerados mejor que codificaciones;
subintervalo mejor que enteros; ...

No usar construcciones de tipos de datos an6nimos:
VAR poligono: ARRAY[1..N] OF ARRAY[(x,y,z)] OF real;

En su lugar, declarar las componentes e irlas identificando una a una:

TYPE
tEjes = (x,y,2);
tPunto = ARRAY[tEjes] OF real;
tNumPunto = 1..N;
tPoligono = ARRAY[tNumPunto] OF tPunto;
VAR

poligono: tPoligono;

3.7 Sobre los comentarios

Los comentarios proporcionan la documentacién del programa. Como norma general, los comentarios
deben ser escuetos pero precisos, aunque pueden ser formales o informales, dependiendo de su finalidad.
Explican los siguientes asuntos en relacién con el programa:

e El cometido del programa, que debe reflejarse al principio del mismo.
e FEl papel que juegan los identificadores, cuando su identificador no baste.

e En relacion con los subprogramas, debe indicarse lo siguiente:

— Para cada parametro, si es de entrada, de salida o ambas cosas.

— Los requisitos de los datos (cuando no baste con su tipo), que se llaman usualmente precondi-
cion del subprograma.

— Su cometido: el resultado que generan las funciones, o el trabajo que desempean los proce-
dimientos. Segun el caso (funcién o procedimiento), el cometido se expresa como el valor
devuelto o la postcondicién, respetivamente.

Asi pues, la especificacion de un subprograma es la siguiente segun el caso:

PROCEDURE P(...); FUNCTION F(...): ...;
{Pre: ...} {Pre: ...}
{Post: ...} {Dev: ...}

e Las instrucciones del programa y de los subprogramas se documentan del siguiente modo:

— El cometido de un fragmento de programa se puede anticipar con un comentario:

BEGIN
{Lectura de datos:}
instrucctones para la lectura de datos
{Calculos:}
instrucciones para los cdlculos
{Presentacién de los resultados:}
tnstrucciones de salida de resultados
END.

Gries resalta el doble papel que desempena este tipo de “instruccién comentada” (command-
comment, en [2]): por una parte, en el refinamiento progresivo de programas, estos comentarios
(seudocodigo) juegan el mismo papel que las instrucciones (codigo) con que se han de refinar
posteriormente; por otro, cuando se lee un programa con una instrucciéon comentada, sélo se
necesita leer el refinamiento si se desea como se ha desarrollado.

— Las propiedades requeridas y garantizadas en los puntos delicados del programa. Ademas de
las precondiciones, postcondiciones, invariantes, ya explicados, conviene a veces anotar otras
propiedades con las que contamos, intercaladas entre las otras instrucciones del programa. Por
ejemplo, entre varias instrucciones de una secuencia,

{a=A, b=B}

a:= atb;

{a=A+B, b=B}

b:= a-b;

{a=A+B, Db=A}

a:= a-b;

{a=B, b=A}
ayudandonos a razonar sobre su verdadero funcionamiento; y también interesa caracterizar la
situacién dentro de una rama de una instruccién de seleccién, como en el ejemplo siguiente.

— En particular, la documentacion tipica de los bucles es su invariante y su expresion cota:
PROCEDURE MCD({in} a, b: integer): integer;
{Pre.: a=A, b=B, a,b>0}

{Dev.: mdximo comin divisor de a y b}
BEGIN {MCD}

{Inv.: a,b>0, mcd(a, b)=mcd(A, B); Cota: max(a, b)}
WHILE a<>b DO

IF a>b THEN
a:= a-b
ELSE {b>a}
b:=b-a;
{a=m.c.d.(a,b) = m.c.d.(A,B)}
MCD:= a
END; {MCD}

3.8 Sangrado

El papel del sangrado es aclarar la estructura del texto de un modo simple y directo; esto es, qué partes
estan subordinadas unas a otras, donde empiezan y terminan las distintas piezas de codigo (declaraciones,
estructuras, modulos), etc. Como el sangrado refleja la sintaxis, las recomendaciones sobre el mismo
atienden a las distintas estructuras sintacticas del lenguaje de programacion.

Aunque no hay unanimidad a proposito de esto, si un par de acuerdos de tipo general: que el cometido
es reflejar la estructura, y que las reglas deben ser consistentes y aplicarse uniformemente.

Veamos algunas recomendaciones generales sobre esto en Pascal.

e En la estructura de programa, las declaraciones estdn subordinadas, situdndose en un nivel méas
interno:

Cabecera
Declaraciones
Cuerpo

Véase el ejemplo global del apéndice A.

e Las instrucciones de una misma secuencia pueden ir con el mismo sangrado:

BEGIN
instruccion
wnstruccions

END

y también pueden agruparse varias, juntas en la misma linea, cuando sean pequenos pasos de una
misma accion:

aux:= X; XI= y; y:= aux

pero no deben colocarse en lineas distintas con distinto nivel de sangrado:

Las instrucciones comentadas estdn al mismo nivel que las ejecutables. El refinamiento de una
instruccién comentada esta subordinado a ella, por lo que se sangra un nivel mas:

instruccion

{Intercambio de las variables a y b:}
aux:= Xx;
X:=y;
y:= aux;

instruccions

e En la instruccion condicional, las dos “ramas” deben estar al mismo nivel; dos posibilidades:

IF condicton THEN IF condicidn
instruccion THEN instruccidng

ELSE ELSE instruccions
wnstruccions

Cuando las ramas sean instrucciones compuestas, los fragmentos anteriores pueden escribirse de
varias formas:

IF condicién THEN BEGIN IF condicidn IF condicién THEN BEGIN
instrucciones THEN BEGIN instrucciones
END {IF} instrucciones END ELSE BEGIN
ELSE BEGIN END {IF} instrucciones
instrucciones ELSE BEGIN END {ELSE}
END {ELSE} instrucciones
END {ELSE}

e En los bucles incondicionales, el cuerpo es la instruccién subordinada:

FOR indice:= wvalor tntctal TO valor final DO
instruccion

aunque, cuando se trata de una secuencia, no hace falta sangrar dos veces:

FOR indice:= wvalor intctal TO valor final DO BEGIN
instruccion ;
nstrucciony;

END; {FOR}

e Los bucles condicionales siguen el mismo criterio:

WHILE cond<cion DO WHILE condtcton DO BEGIN REPEAT

instruccion; instruccion; instruccion;
nstrucciony; instruccions;
END; {WHILE} UNTIL condicion

e Subprogramas: como los programas, las declaraciones estan subordinadas, situdndose en un nivel
mas interno:

Cabecera
Declaraciones
Cuerpo

e Finalmente, no es conveniente un sangrado excesivo, para no rebasar el ancho usual (80 columnas)
del monitor o del papel, trabajando con texto estandar.

3.9 Convenciones sobre el uso de maytsculas y mintisculas

El uso de mayusculas y mintsculas también ayuda a leer y comprender mejor los programas, ya que
permite distinguir mejor diferentes elementos léxicos, como las palabras reservadas o los identificadores
de distinto tipo. En principio, no hay razon alguna para preferir uno u otro convenio, pero si interesa
escoger uno por uniformidad y coherencia.

Un posible conjunto de recomendaciones es el siguiente, que es el que se ha seguido en los ejemplos
de este folleto:

e Usar maytusculas para el comienzo de las constantes y subprogramas.
e Usar mintsculas para el comienzo de las variables.
e Marcar los identificadores de tipos inicialmente con una ‘t’.

e Cuando un identificador conste de varias palabras, se recomienda poner con mayuscula el comienzo
de la segunda, tercera, etc., para facilitar su lectura:

LeerDatos...
ConvertirGradosEnRadianes...

w »

Durante largo tiempo, se ha venido usando también el caracter
distintas palabras en un mismo identificador,

(guién bajo) para separar

Leer_datos...
Convertir_grados_en_radianes...

aunque este caracter no es valido en Pascal estandar.

En cuanto a las palabras reservadas, es costumbre en Pascal distinguirlas, bien con negrita (en los libros)
o subrayandolas (cuando se escribe a mano), para resaltar su papel sintactico, que facilita la lectura. El
inconveniente de estas alternativas es que ninguna de ellas usa texto estandar, de forma que al escribir los
programas en un editor tenemos que pensar en otra posibilidad. Por eso, las palabras reservadas también
se escriben completamente en maytsculas, como estamos haciendo en este manualillo.

3.10 Convenciones a proposito de los tipos abstractos de datos (TADs)

e Todo uso de un TAD debe efectuarse siempre mediante sus operaciones propias.
Posiblemente, el error méas frecuente del usuario de TADs es conocer su implementacion y hacer
uso de ella. Este defecto suprime la abstraccion, lo que contradice evidentemente el objetivo de los
TADs.

e La documentacion necesaria para los modulos de TADs, consiste en lo siguiente:

— Los requisitos de los parametros del TAD.
— La especificacion completa de su comportamiento.
— El coste de cada operacion.

— La implementacién elegida: estructuras de datos e invariante de representacion.

A Apéndice: Ejemplos completos

A.1 Fecha correcta

PROGRAM ComprobacionFechas (input, output);
{Pide repetidamente fechas hasta que se dé una correcta por el teclado}
TYPE
tFecha = RECORD
dia, mes, anyo: integer
END; {tFecha}

FUNCTION EsCorrecta ({in} fecha: tFecha): boolean;
{Pre: No hay requisitos
Dev: True si (dia,mes,anyo) de la fecha dada es una fecha correcta;
False en caso contrario}

FUNCTION NumDias({in} mes, anyo: integer): integer;
{Pre: 1<=mes<=12 anyo>=1800
Dev: el nimero de dias que tiene el mes en el anyo indicado}

FUNCTION EsBisiesto({in} anyo: Integer): Boolean;
{Pre: anyo>=1800
Dev: True si el afio es bisiesto;
Falso en caso contrario}
BEGIN {EsBisiesto}
IF anyo MOD 400 = O THEN
EsBisiesto:= True

ELSE IF anyo MOD 100 = O THEN {anyo MOD 100=0 y anyo MOD 400<>0}

EsBisiesto:= False
ELSE {anyo MOD 100<>0 y anyo MOD 400<>0}
EsBisiesto:= anyo MOD 4 = 0
END; {EsBisiesto}

BEGIN {NumDias}

CASE mes OF
1,3,5,7,8,10,12: NumDias:= 31;
4,6,9,11 : NumDias:= 30;
2 : IF EsBisiesto(anyo) THEN
NumDias:= 29
ELSE
NumDias:= 28
END {CASE}

END; {NumDias}

BEGIN {EsCorrecta}
WITH fecha DO
esCorrecta:= (anyo>=1800) AND (mes>=1) AND (mes<=12) AND
(dia>=1) AND (dia<=NumDias (mes,anyo))
END; {EsCorrecta}

VAR
fecha: tFecha;

BEGIN {ComprobacionFechas}
REPEAT
WriteLn(’Dame la fecha: ?);
WITH fecha DO
ReadLn(dia,mes,anyo)
UNTIL EsCorrecta(fecha)
END. {ComprobacionFechas}

10

A.2 Fecha correcta

El siguiente programa es correcto desde el punto de vista del funcionamiento, e incluso estd bien
documentado. Sin embargo, se presenta como programa inaceptable por razones obvias:

PROGRAM ComprobacionFechas (input, output); {Pide repetidamente fechas
hasta que se dé una correcta por el teclado} TYPE tFecha = RECORD

dia, mes, anyo: integer END; {tFechal} FUNCTION EsCorrecta ({in}

fecha: tFecha): boolean; {Pre: No hay requisitos Dev True si (dia,
mes,anyo) de la fecha dada es una fecha correcta; False en caso
contrario} FUNCTION NumDias({in} mes,anyo: integer):integer; {Pre:
1<=mes<=12 anyo>=1800 Dev: el numero de dias que tiene el mes en el
anyo indicado} FUNCTION EsBisiesto({in} anyo: Integer): Boolean;

{Pre: anyo>=1800 Post: La funcidén devuelve True si el afio es bisiesto;
Falso en caso contrario} BEGIN {EsBisiesto} IF anyo MOD 400 = O THEN
EsBisiesto := True ELSE IF anyo MOD 100 = O THEN {anyo MOD 100=0 y anyo
MOD 400<>0} EsBisiesto := False ELSE {anyo MOD 100<>0 y anyo MOD 400<>0}
EsBisiesto := anyo MOD 4 = O END; {EsBisiesto} BEGIN {NumDias} CASE

mes OF 1,3,5,7,8,10,12 : NumDias := 31; 4,6,9,11 : NumDias := 30; 2 :
IF EsBisiesto(anyo) THEN NumDias := 29 ELSE NumDias := 28 END {CASE}
END; {NumDias} BEGIN {EsCorrecta} WITH fecha DO esCorrecta :=
(anyo>=1800) AND (mes>=1) AND (mes<=12) AND (dia>=1) AND

(dia<=NumDias (mes,anyo)) END; {EsCorrecta} VAR fecha: tFecha; BEGIN
{ComprobacionFechas} REPEAT WriteLn(’Dame la fecha: ’); WITH fecha DO
Readln(dia,mes,anyo) UNTIL EsCorrecta(fecha) END. {ComprobacionFechas}

A.3 Minimo elemento de un conjunto

El siguiente fragmento de programa ejemplifica el uso de propiedades, formalmente, para especificar
un subprograma y un bucle.

CONST
Min= 1; Max= 50;

TYPE
tNaturales= Min..Max;
tConjuntoNaturales= SET OF tNaturales;

FUNCTION MinimoElemento({in} conjunto: tConjuntoNaturales): tNaturales;
{Pre: conjunto # &
Dev: n € conjunto.(Vk < n.k & conjunto)}
VAR
n: tNaturales;
BEGIN {MinimoElemento}
n:= Min;
{Inv: VEk <n.k ¢ conjuntoA Ik > n.k € conjunto. Cota: Max-n}
WHILE NOT (n IN conjunto) DO
n:= nt+l;
{n € conjunto A (Vk < n.k & conjunto)}
MinimoElemento:= n
END; {MinimoElemento}

Las propiedades anteriores pueden expresarse también informalmente, aunque lo importante es que

se haga con la misma precisién y claridad:

CONST
Min= 1; Max= 50;

TYPE
tNaturales = Min..Max;
tConjuntoNaturales = SET OF tNaturales;

FUNCTION MinimoElemento({in} conjunto: tConjuntoNaturales): tNaturales;
{Pre: conjunto no es vacio
Dev: el minimo elemento de conjunto}
VAR
n: tNaturales;
BEGIN {MinimoElemento}
n:= Min;
{Inv: ningin elemento menor que n estad en el conjunto,
y a partir de n si hay algin elemento en conjunto}
WHILE NOT (n IN conjunto) DO
n:= nt+l;
{n es el minimo elemento del conjunto}
MinimoElemento:= n
End; {MinimoElemento}



